


Artificial Intelligence (CS308)

Assignment - 3

U19CS012

1.) Consider the Following Story.

"I married a **widow** (call her W) who has a **grown-up daughter** (D). My **father** (F), who visited us quite often, fell in love with my **step-daughter** and married her. Hence my father became my **son-in-law** and my **step-daughter** became my **mother**. Some months later, **my wife** gave birth to a **son** (S1), who became the **brother-in-law** of my **father**, as well as my **uncle**. The wife of my father (i.e) my **step-daughter** also had a son (S2)."



As a husband of my grandma...

I AM MY OWN GRANDPA

- Using Prolog, Create a List of Facts that represents the situation in the above Story.
- Add **Rules** defining the family relationships (such as father-in-law) described in the story.

PROLOG Code

```
% [U19CS012] BHAGYA VINOD RANA

% Facts

% male(person) - person is male
male(f).
male(i).
male(s1).
male(s2).

% female(person) - person is female
female(w).
female(d).

% husband(h,w) - h is the husband of w
husband(i,w).
husband(f,d).

% father(f,s) - f is the father of s
father(i,s1).
father(f,s2).
father(f,i).

% mother(m,s) - m is the mother of s
mother(d,s2).
mother(w,s1).
mother(w,d).

% Relationships [Rules]

% Wife Relation [X is Wife of Y, if Y is the Husband of X]
wife(X, Y) :- husband(Y, X).

% For Married Condition, X should be either Husband
married(X, Y) :- husband(X, Y).
married(X, Y) :- wife(X, Y).

% biological parent
bio_parent(X, Y) :- father(X, Y).
bio_parent(X, Y) :- mother(X, Y).

% daughter Relationship
daughter(X, Y) :- female(X), bio_parent(Y, X).

% son
son(X, Y) :- male(X), bio_parent(Y,X).

% step parent -> someone which your father/mother marries after
```

```

% old relationship ends
step_parent(X, Z) :- married(X, Y), bio_parent(Y, Z), \+ bio_parent(X, Z).

step_father(X, Z):- step_parent(X, Z), male(X).
step_mother(X, Z):- step_parent(X, Z), female(X).
step_daughter(X, Y):- female(X), step_parent(Y, X).
step_son(X, Y):- male(X), step_parent(Y,X).

parent(X, Y) :- step_parent(X, Y).
parent(X, Y) :- bio_parent(X, Y).

% the father of one's husband or wife.
father_in_law(X, Y):- married(Y, Z), parent(X, Z), \+ parent(X, Y), male(X).
mother_in_law(X, Y):- married(Y, Z), parent(X, Z), \+ parent(X, Y), female(X).
parent_in_law(X, Y) :- father_in_law(X, Y). parent_in_law(X, Y) :- mother_in_law(X, Y).
son_in_law(X, Y):- parent_in_law(Y, X), male(X).
daughter_in_law(X, Y):- parent_in_law(Y, X), female(X).

% grandfather relationship
grandfather(X, Z) :- parent(X, Y), parent(Y, Z), male(X).

sibling(X, Y):- parent(Z, X), parent(Z, Y), X \= Y.
dist_sibling(X, Y) :- distinct(sibling(X, Y)).

% brother in law - the brother of one's wife or husband.
brother_in_law(X, Z):- married(Y, Z), sibling(X, Y), male(X).
sister_in_law(X, Z):- married(Y, Z), dist_sibling(X, Y), female(X).

% uncle
uncle(X, Y):- dist_sibling(X, Z), parent(Z, Y), male(X).

```

So, Let's **Execute the Prolog** file and **Check** if the facts and relationships are defined correctly.

"i" married a widow "w".

```

?- married(i,w).
true .

?- husband(X,w).
X = i.

?- wife(X,i).
X = w.

```

"w" has a daughter "d".

```
?- daughter(d,w).  
true.
```

```
?- mother(w,d).  
true.
```

"d" is step-daughter of "i".

```
?- step_daughter(d,i).  
true .
```

```
?- step_father(i,d).  
true .
```

"d" married to i's father "f".

```
?- married(f,d).  
true .
```

Hence, "f" became son-in-law of "i".

```
?- son_in_law(f,i).  
true .
```

and i's step-daughter "d" became his step-mother.

```
?- step_mother(d,i).  
true .
```

"i" and "w" gave birth to son "s1".

```
?- bio_parent(W,s1).  
W = i ;  
W = w.
```

hence, "s1" became brother-in-law of "f".

```
?- brother_in_law(s1,f).  
true .
```

and "s1" became uncle of "i".

```
?- uncle(s1,i).  
true .
```

"d" and "f" had a son "s2".

```
?- parent(X,s2).  
X = f ;  
X = d.
```

Show how a Prolog system would use your program to Prove the Goal

"I am my own grandfather"

```
?- grandfather(i,i).  
true .
```

I am my own grandfather!

Show how a **Prolog** system would use your program to **prove** the goal "I am my own grandfather".

- ① Using Predicate Calculus, we have created set of expression that represents the situation in above story.
- ② Used Expressions defining basic family relationship like father-in-law.
- ③ Using Modus Ponens on this system to prove that
"I am my own grandfather"

- (1) $\text{father}(X, Z) \wedge \text{parent}(Z, Y) \rightarrow \text{grandfather}(X, Y)$
- (2) $\text{father}(X, Y) \wedge \text{marry}(Y, Z) \rightarrow \text{father-in-law}(X, Z)$
- (3) $\text{marry}(X, Y) \wedge \text{mother}(X, Z) \rightarrow \text{step-father}(Y, Z)$
- (4) $\text{father-in-law}(X, Y) \vee \text{step-father}(X, Y) \rightarrow \text{father}(X, Y)$
- (5) $\text{father}(X, Y) \vee \text{mother}(X, Y) \rightarrow \text{parent}(X, Y)$
- (6) $\text{marry}(W, I)$
- (7) $\text{mother}(W, D)$
- (8) $\text{father}(F, I)$
- (9) $\text{marry}(D, F)$
- (10) $\text{step-father}(I, D)$ from (3), (6), (7) and modus ponens
- (11) $\text{father}(I, D)$ from (4), (10) and modus ponens
- (12) $\text{father-in-law}(I, F)$ from (2), (9), (11) and modus ponens.
- (13) $\text{father}(I, F)$ from (4), (12) and modus ponens.
- (14) $\text{parent}(F, I)$ from (8), (13) and modus ponens.
- (15) $\text{grandfather}(I, I)$ from (1), (13), (14) and modus ponens.

\therefore I am my own grandfather.

Modus Ponens

If P implies Q - P is true.

Therefore, Q must also be true.

SUBMITTED BY: U19CS012

BHAGYA VINOD RANA